
BASIC TOWER 
 
Basic Tower es un demake tributo al innovador juego OldTower del grupo 
RetroSouls. 
Está programado en Basic Puro para el Spectrum 48kb. 
 
Nos movemos a través de 16 fases multicolor mediante un scroll vertical, sin 
parpadeos, recogiendo las monedas que nos permiten abrir la salida hasta el 
siguiente piso de la torre. 
 
Hay que tener cuidado con los fantasmas y los disparos de las gárgolas, que nos 
quitarán una vida. 
 
Las fases tienen seis conocidas melodías, a través de Beep in-game. 
El juego es multicarga, una para las fases de 1 a 7, otra de 8 a 13 y la ultima de 
14 a 16.  
 
Las teclas son: 
OPQA para movernos. El personaje no puede cambiar de dirección hasta que no 
se detenga chocando  
con un muro. 
M para cambiar de personaje. 
S para desactivar el sonido. 
 
 
Técnica 
 
Las animaciones del personaje, objetos y escenario se consiguen usando 4 
bancos de UDGs, alternándolos en cada repeticián del bucle principal. 
 
Para conseguir un scroll *rápido* en Basic, cada vez que se sube o baja, solo 
imprime las partes del escenario que hayan cambiado con respecto al frame 
anterior. 
 
Para ello, mediante un preprocesado del mapa de cada fase, se obtienen dos 
cadenas de unas 60 filas, con la pantalla moviendo hacia abajo, y otra moviendo 
hacia arriba, 
Se calculan el inicio y la longitud de cada pantalla de 20 filas, y a la hora de 
imprimirla, se le pasan esos dos valores a la cadena z$ y se pinta con DefAdd. 
 
Cada fase ocupa 3 Kb, repartidos de este modo: 
1,5 kb para las cadenas de 60 filas del mapa subiendo o bajando 
0,3 kb con la posicion de inicio y el tamaño de cada uno de los 40 frames de cada 
fase 
1 kb para guardar la direccion de los objetos con los que se puede interactuar en 
el mapa. 
0,2 kb datos de los fantasmas y gargolas de la fase.  
0,1 kb melodia de esa fase 

 
 



Código 
 

BUCLE PRINCIPAL 
 
Mover Fantasma, si está visible en este frame 
3 LET h=h+(-19 OR h<10):  
   IF PEEK (41984+c) THEN  

POKE 38979,ABS h:  
LET i$=f$: LET y$=f$:  
PRINT INK 7;AT PEEK (41984+c)-1,1;f$;AT 0,0; 

 
Mover disparo de Gargolas, si están visibles en este frame 
6 IF PEEK (41983+c) THEN  

POKE 39006,PEEK (39061+h):  
LET c$=g$: LET k$=g$:  
PRINT INK 7;AT PEEK (41982+c+k)-1,2;g$;AT 0,0;:  
FOR e=1 TO NOT h OR h=10:  

POKE 39006,26:  
PRINT z$; INK 7;AT PEEK (41982+c+k)-1,2;g$;AT 0,0;:  
LET c$=g$: LET k$=g$: LET k=NOT k:  
POKE 23563,18 AND k:  

NEXT e 
 

Tocar una nota de la melodía 
9 IF PEEK 23672>12 THEN  

LET s=s+1-(95 AND s=42239):  
POKE 23672,0:  
IF PEEK s THEN  

BEEP .02,PEEK s 
 
Mover personaje a izq o drch, hasta topar con un muro 
11 IF 1=ABS j THEN  

LET e=ATTR (a,b+j):  
FOR u=1 TO PEEK (39028+e):  

POKE 23676,PEEK (39105+h):  
LET b=b+j: POKE 65332,b:  
PRINT e$;:  
POKE 65323,e: POKE 65326,b:  
GO TO PEEK (39329+e):  

NEXT u:  
LET d=i(w+PEEK (39464+a*12+c*3+b)):  
POKE d,7: POKE d+1,PEEK 65333: LET j=0:  
GO TO 50 
 

Mover personaje hacia arriba o abajo, hasta topar con un muro 
20 IF j THEN  

LET e=ATTR (a+SGN j,b):  
FOR u=1 TO PEEK (39028+e):  

GO TO 142+j:  
NEXT u:  
LET d=i(w+PEEK (39464+a*12+c*3+b)):  



POKE d,7: POKE d+1,PEEK 65333: LET j=0 
 
Mover la cámara hasta centrar al personaje en las filas de mitad de pantalla 
96 IF a<9 THEN  

POKE 38961,c:  
LET p$=q$: LET c=c-4: LET a=a+1:  
POKE 65325,a: POKE 65331,a 

97 IF a>11 THEN  
LET c=c+4: POKE 38970,c:  
LET p$=r$: LET a=a-1:  
POKE 65325,a: POKE 65331,a 

110 PRINT z$;e$;AT 7,20; INK 6;"\f\  ";p+t;AT 0,0;:  
       POKE 23676,PEEK (39105+h):  ‘Actualiza animaciones 
       GO TO 225 AND INKEY$>""  ‘Si hay tecla pulsada, salta 
 

FIN DEL BUCLE PRINCIPAL 
 
*********************************************************************************** 
Scroll vertical arriba 
 140 IF a>3 THEN  

LET a=a-1: POKE 65331,a:  
PRINT e$;:  
POKE 65323,e: POKE 65325,a:  
GO TO PEEK (39074+e) 

 141 POKE 38961,c:  
LET p$=q$: LET c=c-4:  

    PRINT z$;e$;: POKE 65323,e:  
GO TO PEEK (39074+e) 

Scroll vertical abajo 
 161 IF a<17 THEN  

LET a=a+1: POKE 65331,a:  
PRINT e$;:  
POKE 65323,e: POKE 65325,a:  
GO TO PEEK (39074+e) 

 167 LET c=c+4: POKE 38970,c:  
LET p$=r$:  
PRINT z$;e$;: POKE 65323,e:  
GO TO PEEK (39074+e) 

 
*********************************************************************************** 
Detección de colisiones por ATTRibutos 
Amarillo  
206 LET e=PEEK (39464+a*12+c*3+b):  

POKE i(e),0: POKE y(e),0:  
POKE 65323,0: LET p=p+1:  
IF p=PEEK 39422 THEN  

FOR e=39414 TO 39417:  
IF PEEK e<95 THEN  

POKE i(PEEK e)+1, CODE "\q":  
POKE y(PEEK e)+1, CODE "\q":  
NEXT e 



Cyan 
 215 GO TO 3 
  
 
Blanco 
217 LET e=a+c*.25:  

POKE 65323,7 AND e<>PEEK 39413 AND e<>PEEK 39411 AND 
e<>PEEK 39412:  

POKE 65327,148:  
GO TO (1998 AND p=PEEK 39422 AND PEEK 65323) 

 
*********************************************************************************** 
TECLADO 
Q - arriba 
 225 IF IN 64510=190 THEN  

LET e=ATTR (a-1,b): 
IF PEEK (39028+e) THEN   ’colision? 

POKE 65333,147: LET j=-2:  
LET w=0: POKE d,0:  
GO TO 140 

A - abajo 
 227 IF IN 65022=190 THEN  

LET e=ATTR (a+1,b):  
IF PEEK (39028+e) THEN  ’colision? 

POKE 65333,144: LET j=2:  
LET w=96: POKE d,0:  
GO TO 160 

O - izquierda 
 230 IF IN 57342=189 THEN  

LET e=ATTR (a,b-1):  
IF PEEK (39028+e) THEN  ’colision? 

LET j=-1: LET b=b-1:  
POKE 65332,b:  
POKE 65333,145+(2 AND PEEK 65333> CODE "\b"):  
PRINT e$;: POKE 65323,e:  
POKE 65326,b: POKE d,0:  
GO TO PEEK (39329+e) 

P - derecha 
 232 IF IN 57342=190 THEN  

LET e=ATTR (a,b+1):  
IF PEEK (39028+e) THEN  ’colision? 
LET j=1: LET b=b+1:  
POKE 65332,b:  
POKE 65333,144+(2 AND PEEK 65333> CODE "\b"):  
PRINT e$;: POKE 65323,e:  
POKE 65326,b: POKE d,0:  
GO TO PEEK (39329+e) 

M – cambia de personaje 
 234 IF INKEY$="M" THEN  

IF m2 THEN  
GO SUB 1501:  



LET j=0:  
GO TO 3 

’R – reintentar esta fase 
 236 IF INKEY$="R" THEN  

IF fa THEN  
GO TO 247 

’S – apagar sonido 
 237 IF INKEY$="S" THEN  

POKE 24201,234 
 

 240 GO TO 2 
*********************************************************************************** 
Otra detección de colisiones por ATTRibutos menos habitual 
Azul 
 241 POKE 65327, CODE "\l": GO TO 3 
Magenta 
 243 LET e=PEEK (39464+a*12+c*3+b): POKE i(e)-2,1: POKE y(e)-2,1: POKE 
i(e)+1,150: POKE y(e)+1,150: POKE 65327,150: GO TO 3 
Verde  
244 POKE 65327,156: GO TO 3 
Blanco 
247 GO TO 1900 
Cyan y Azul 
248 LET e=PEEK (39464+a*12+c*3+b):  

POKE i(e),0: POKE y(e),0:  
POKE i(e)-2,0: POKE y(e)-2,0:  
IF ABS j=1 THEN  

POKE 65323,0: POKE 65332,b:  
LET j=0: PRINT e$;:  
POKE 65326,b:  
GO TO 50 

 249 POKE 65323,0: POKE 65331,a:  
LET j=0: PRINT e$;:  
POKE 65325,a:  
GO TO 50 

250 IF ABS j=1 THEN  
LET b=b-j: POKE 65323,0:  
POKE 65332,b: LET j=0:  
PRINT e$;: POKE 65326,b:  
GO TO 50 

 
*********************************************************************************** 
Después de cambio de personaje, mover la cámara hasta el nuevo 
 960 IF a<1 THEN  

POKE 38961,c: LET p$=q$:  
LET c=c-4: LET a=a+1:  
POKE 65331,a: POKE 65325,a:  
PRINT z$;AT 0,0;:  
GO TO 960 

 970 IF a>19 THEN  
LET c=c+4: POKE 38970,c:  



LET p$=r$: LET a=a-1:  
POKE 65331,a: POKE 65325,a:  
PRINT z$;AT 0,0;:  
GO TO 970 

 990 RETURN  
 
*********************************************************************************** 
Intercambio de personaje 
1500 POKE d,0: POKE d+1,58+(102 AND PEEK 65323): LET d=i((96 AND 
NOT w)+PEEK (39464+a*12+c*3+b)): POKE d,0: POKE d+1,58+(102 AND 
PEEK 65323): GO TO 1502: REM incr blanco como salida o verde 
1501 POKE d,4: POKE d+1,58+(102 AND PEEK 65323): LET d=i((96 AND 
NOT w)+PEEK (39464+a*12+c*3+b)): POKE d,4: POKE d+1,58+(102 AND 
PEEK 65323) 
1502 IF NOT pr THEN LET pr=2: LET m0=a+c*.25: LET b0=b: LET w0=w: LET 
a=m2-c*.25: LET b=b2: LET w=w2: LET e=i((96 AND NOT w)+PEEK 
(39464+m2*12+b)): LET d=i(w+PEEK (39464+m2*12+b)): GO TO 1520 
1505 LET pr=0: LET m2=a+c*.25: LET b2=b: LET w2=w: LET a=m0-c*.25: LET 
b=b0: LET w=w0: LET e=i((96 AND NOT w)+PEEK (39464+m0*12+b)): LET 
d=i(w+PEEK (39464+m0*12+b)) 
1520 POKE e,0: POKE d,7: POKE d+1,144: GO SUB 960: POKE 65325,a: 
POKE 65326,b: POKE 65327,32: POKE 65329,7: POKE 65331,a: POKE 
65332,b: POKE 65333,PEEK (d+1): RETURN : REM desincr verde y actualizo 
e$.  
 
*********************************************************************************** 
Muerte 
1900 LET l=l-1: LET p=0:  

LET e=0: POKE d,5:  
FOR u=0 TO 9:  

PRINT AT 0,0;z$;AT 7,20; INK 6;"\f";AT a,b; INK 
5;"\a:\b::\a\a::\b"(u+1);:  

POKE 23676,(PEEK 23676 AND PEEK 23676>252)-1:  
BEEP .12 AND PEEK (64835+u),PEEK (64835+u):  

NEXT u:  
PRINT AT 16,20; INK 7;"\a\a\a\a"( TO l); INK 1;"\c\c\c\c"( TO 4-l) 

1901 LET u=(u+1 AND u<9):  
PRINT AT 0,0;z$;AT 7,20; INK 6;"\f";AT a,b; INK 5;"\a:\b::\a\a::\b"(u+1);:  
POKE 23676,(PEEK 23676 AND PEEK 23676>252)-1:  
IF INKEY$="" AND e<15 THEN  

LET e=e+1: GO TO 1901 
1902 IF l THEN  

GO SUB 7300:  
GO SUB 7600:  
GO TO 2006:  

 
*********************************************************************************** 
Reiniciar 
1905 POKE 23607,89: PRINT AT 20,10; FLASH 1;"GAME\  OVER": POKE 
23607,150 
1907 PRINT AT 0,0;z$;AT 7,20; INK 6;"\f";:  



POKE 23676,(PEEK 23676 AND PEEK 23676>252)-1:  
IF INKEY$="" THEN  

GO TO 1907 
1910 LET fa=NOT PI: LET t=NOT PI: LET l=4: GO TO 2000 
1999 IF m2 THEN  

LET j=0:  
GO SUB 1500:  
LET m2=0:  
GO TO 3: REM si hay espejo, que salgan los dos. 

*********************************************************************************** 
Paso de fase 
2000 LET t=t AND fa: LET p=p AND fa:  

LET fa=1+(fa AND fa<16) 
2001 GO SUB 7220:  

LET e=1: LET c=0: LET j=1: LET t=t+p: LET s=64800:  
FOR u=40064+1 TO 40064+PEEK 39423:  

LET i(e)=256*PEEK (288+u)+PEEK (192+u):  
LET i(e+96)=256*PEEK (96+u)+PEEK u:  
LET y(e)=i(e+96): LET e=e+1:  
BEEP .02 AND PEEK s,PEEK s:  
LET s=s+1: POKE 23672,0:  
IF s=64834 THEN LET s=64800 

2002 PRINT AT 20,c; INK 6;"\j\f\j"; INK 7;AT 
a,b;"\b\b\a\a\b\b\a\a\a\b\b\b\a\a\a\a\b\b\b\b\a\b\a\a\b\b\a\a\a\b\b\b\a\a\a\a\b\b\b\b
"(c+1):  

POKE 23676,(PEEK 23676 AND PEEK 23676>252)-1:  
LET c=c+j: IF NOT c OR c>25 THEN LET j=-j 

2005 NEXT u: GO SUB 7600 
2006 IF INKEY$>"" THEN GO TO 2006 
2010 POKE 23560,NOT PI: GO TO 2 
 
*********************************************************************************** 
Inicio del programa, declaracion de variables 
7100 CLEAR 38783: RESTORE 7100: PAPER NOT PI: BORDER NOT PI: 
POKE 23658,8:POKE 23561,SGN PI: POKE 23562,SGN PI: LET e=14: LET 
a=SGN PI: LET c=SGN PI: LET b=2: LET j=NOT PI: LET u=NOT PI: LET 
h=NOT PI: LET d=VAL "65368": DIM i(192): DIM y(96): LET w=96: LET 
s=64800: LET p=NOT PI: LET k=SGN PI: LET t=NOT PI: LET l=4: LET fa=NOT 
PI:  
7200 LET fa=-1: GO TO 2000 
 
*********************************************************************************** 
Cargar fases de cinta (multicarga) 
7220 POKE 23607,89:  

IF fa=1 THEN IF PEEK 39410>7 THEN  
PRINT AT 20,9; FLASH 1; INK 2;"START\  THE\  TAPE": POKE 

23739,111:  
LOAD "mapas0a6" CODE VAL "39168",VAL "21504": REM a menos que 

ya este en memoria!! 
7222 IF fa=7 THEN IF PEEK 39410<>7 THEN  



PRINT AT 20,9; FLASH 1; INK 2;"START\  THE\  TAPE": POKE 
23739,111:  

LOAD "mapas7a13" CODE VAL "39168",VAL "21504" 
7225 IF fa=14 THEN IF PEEK 39410<>14 THEN  

PRINT AT 20,9; FLASH 1; INK 2;"START\  THE\  TAPE": POKE 
23739,111:  

LOAD "mapas14a18" CODE VAL "39168",VAL "21504": 
 
Cargar fase a memoria con DefAdd 
7330 POKE 23563,238:  

POKE 39155,153+12*(fa-(7 AND fa>6)-(7 AND fa>13)):  
LET d$=o$: POKE 23563,18:  
POKE 23607,89+(61 AND fa):  
POKE 23739,244: 
RETURN 

 
*********************************************************************************** 
Inicializacion de variables 
7600 LET c=4: LET a=PEEK 39418: LET b=PEEK 39419: LET j=NOT PI: LET 
h=10: LET d=65368: LET w=96: LET t=t AND fa: LET p=0: LET s=42145: LET 
pr=NOT PI: LET m2=PEEK 39420: POKE 39032,NOT PEEK 39420: POKE 
39033,fa<10: RESTORE 7600: FOR u=65322 TO 65336: READ e: POKE u,e: 
NEXT u: LET e=1: DATA 16,NOT PI,22,20,b,32,16,7,22,a,b, CODE "\a",22,NOT 
PI,NOT PI: REM e$ 
Datos del Fantasma 
7620 POKE 38979,68-h: POKE 38981,NOT PI:  

LET e=PEEK (39465+12*PEEK 39413) AND PEEK 39413:  
IF e THEN POKE 38981,16:  
RANDOMIZE y(e)+1:  
POKE 38988,PEEK 23670: POKE 38989,PEEK 23671:  
RANDOMIZE i(e)+1:  
POKE 38997,PEEK 23670: POKE 38998,PEEK 23671 

Datos de Gargolas 
7660 POKE 38912+90+6,0:  

LET e=PEEK (39465+12*PEEK 39411) AND PEEK 39411:  
IF e THEN  

POKE 39008,14: POKE 39006,PEEK (65055+h):  
RANDOMIZE y(e)+4:  
POKE 39015,PEEK 23670: POKE 39016,PEEK 23671: 
RANDOMIZE i(e)+4:  
POKE 39024,PEEK 23670: POKE 39025,PEEK 23671 

Datos de Gargolas 
7661 POKE 38918,NOT PI:  

LET u=PEEK (39465+12*PEEK 39412) AND PEEK 39412:  
IF NOT u THEN  

LET u=e: POKE 39412,PEEK 39411 
7662 IF u THEN  

POKE 38918,14:  
RANDOMIZE y(u)+4:  
POKE 38916,PEEK 23670: POKE 38917,PEEK 23671:  
RANDOMIZE i(u)+4:  



POKE 38925,PEEK 23670: POKE 38926,PEEK 23671 
 

7700 INK 2+(3 AND fa>9): CLS : FOR u=19 TO NOT PI STEP -1: PRINT AT 
u,0;"\o";AT u,13;"\o";: NEXT u: POKE 23563,18: POKE 23564,152: LET 
a=PEEK 39418: PRINT AT 16,20; INK 7;"\a\a\a\a"( TO l); INK 1;"\c\c\c\c"( TO 4-
l);AT 0,0;: LET d=y(PEEK (39464+a*12+b)): POKE d,7: POKE d+1, CODE "\a": 
IF m2 THEN LET b2=PEEK 39421: LET w2=w: LET e=i(PEEK 
(39464+m2*12+b2)): POKE e,4: POKE e+1, CODE ":": LET e=y(PEEK 
(39464+m2*12+b2)): POKE e,4: POKE e+1, CODE ":": REM incrustar Doble: 
7710 LET a=a-1: GO SUB 960: POKE 23560,NOT PI: RETURN 


